

## Chapter 5

# Applications of Differentiation

Maple is used in this chapter to solve some classic problems associated with differentiation; namely, related rate problems, graphical analysis problems and max/min problems.

### 5.1 Related Rates

In many situations, two or more quantities are related by some formula and are changing with time. In a related rates problem, the goal is to determine how fast one of the quantities is changing when the rates of change of the other quantities are known.

The procedure for solving this type of problem can be outlined as follows:

1. Determine the quantities that are changing with time (or the parameter).
2. Find an equation that relates the quantities (this may involve known formulas from physics or geometry).
3. Enter the equation into Maple, writing the quantities as functions of time.
4. Take the implicit derivative of the equation with respect to time.
5. Solve the resulting equation for the desired derivative.
6. Find numeric values for each of the quantities in the resulting solution, possibly by using the equation from step 3, and substitute them into the result.

The following is a typical example.

EXAMPLE: For a concave lens with focal length 30 cm, the following formula from optics expresses the relationship between the distance  $u$  from an object to the lens and the distance  $v$  from the lens to its image.

$$\frac{1}{30} = \frac{1}{u} + \frac{1}{v}$$

Suppose an object is moving towards the lens at a rate of 3 cm/sec. Find the rate at which the image is receding from the lens when the object is 90 cm away.

SOLUTION: The two quantities that change with time are the two distances,  $u$  and  $v$ . The equation that relates them is the optics formula given in the problem. This equation is entered into Maple, using  $u(t)$  and  $v(t)$  to signify that the distances are time-dependent.

```
> eq:=1/30=1/u(t)+1/v(t);
```

$$eq := \frac{1}{30} = \frac{1}{u(t)} + \frac{1}{v(t)}$$

This equation is differentiated with respect to  $t$ .

```
> deq:=diff(eq,t);
```

$$deq := 0 = -\frac{\frac{d}{dt} u(t)}{u(t)^2} - \frac{\frac{d}{dt} v(t)}{v(t)^2}$$

The resulting equation is solved for  $v'(t)$ .

```
> vrate:=solve(deq,diff(v(t),t));
```

$$vrate := -\frac{(\frac{d}{dt} u(t)) v(t)^2}{u(t)^2}$$

The numerical value of  $v$  is found.

```
> u0:=90; urate0:=-3;
```

$$\begin{aligned} u0 &:= 90 \\ urate0 &:= -3 \end{aligned}$$

```
> solve(eq,v(t)); v0:=subs(u(t)=u0,%);
```

$$\begin{aligned} &\frac{30 u(t)}{u(t) - 30} \\ v0 &:= 45 \end{aligned}$$

The answer is found by substituting these values into `vrate`.

```
> vrate0:=subs({u(t)=u0, diff(u(t),t)=urate0, v(t)=v0}, vrate);
```

$$vrate0 := \frac{3}{4}$$

So the image is moving away at  $\frac{3}{4}$  cm/sec.

## 5.2 Local Extrema

In this section, you will learn how to use the `plot`, `diff`, `solve` and `fsolve` commands to find local maxima and minima (also called local extrema) of differentiable functions. If  $f$  is a differentiable function on an open interval, then its derivative must vanish (i.e.,  $f' = 0$ ) at each local maximum or minimum. So the strategy is to first plot the function to get an approximate idea of the location of the maxima and minima. Then use the `solve` or `fsolve` command to find the solutions of the equation  $f' = 0$ .


EXAMPLE: Plot the expression  $f = x^3 + 0.2x^2 - x$  and find the location of the local maxima and minima.

SOLUTION: Define  $f$  and set  $Df$  equal to its derivative.

```
> f:=x^3+0.2*x^2-x;
      f := x3 + 0.2x2 - x
> Df:=diff(f,x);
      Df := 3x2 + 0.4x - 1
```

Plot  $f$  for  $-2 \leq x \leq 2$ .

```
> plot(f, x=-2..2);
```



Observe there is a local maximum between  $-1$  and  $0$  and a local minimum between  $0$  and  $1$ . To find the precise locations, solve the equation  $Df = 0$ . The local maximum is between  $-1$  and  $0$  and is labeled `xmax`:

```
> xmax:=fsolve(Df=0,x=-1..0);
      xmax := -0.6478531925
```

The local minimum is between  $0$  and  $1$  and is labeled `xmin`.

```
> xmin:=fsolve(Df=0,x=0..1);
      xmin := 0.5145198591
```

Inserting these values back into the original expression  $f$  will give the corresponding  $y$ -coordinates of the local maximum and local minimum.

```
> ymax:=subs(x=xmax,f); ymin:=subs(x=xmin,f);
      ymax := 0.4598830456
      ymin := -0.3253645270
```

Therefore, the local maximum is at the point  $(-0.6478531925, 0.4598830456)$  and the local minimum is at  $(0.5145198591, -0.3253645270)$ .

NOTE: Above  $f$  was defined as a function and we used `subs`. If instead,  $f$  is defined as an arrow-defined function,

```
> f:=x->x^3+0.2*x^2-x;
```

$$f := x \rightarrow x^3 + 0.2x^2 - x$$

then the  $x$ -coordinate of the local maximum can be found from

```
> xmax:=fsolve(D(f)(x)=0,x=-1..0);
```

$$xmax := -0.6478531925$$

and the corresponding  $y$ -coordinate can then be obtained from

```
> ymax:=f(xmax);
```

$$ymax := 0.4598830456$$

Of course, the above example could have been done by hand. However, the example in the next section, and many of the exercises are too complicated for hand computation.

### 5.3 Graphical Analysis

In this section, we consider, in detail, the graph of the expression:

$$f := \frac{e^x}{x^3 + x - 1 + 0.2e^x}$$

The goal is to produce an accurate graph, locating horizontal and vertical asymptotes, local extrema and inflection points.


First, enter the expression into Maple with  $e^x$  entered as `exp(x)`.

```
> f:=exp(x)/(x^3+x-1+0.2*exp(x));
```

$$f := \frac{e^x}{x^3 + x - 1 + 0.2e^x}$$

Plot the graph over the interval  $-5 \leq x \leq 15$ .

```
> plot(f, x=-5..15, y=-10..10, discontin=true);
```



The  $y$ -range is restricted to  $-10 \leq y \leq 10$  to obtain a reasonable plot and a line at the vertical asymptote has been eliminated by including the `discontin=true` option. From the graph, it appears that there is a vertical asymptote between  $x = 0$  and  $x = 1$ . Its location can be pinpointed by locating the root of the

denominator of  $f$ . In Maple, the denominator of an expression  $f$  is `denom(f)`. Using `denom(f)` saves typing in cases where the denominator is complicated.

```
> vasymp:=fsolve(denom(f)=0,x=0..1);
      vasymp := 0.5213890506
```

We find horizontal asymptotes by taking limits

```
> Limit(f,x=-infinity); value(%);
      
$$\lim_{x \rightarrow (-\infty)} \frac{e^x}{x^3 + x - 1 + 0.2 e^x}$$

      0.
```

```
> Limit(f,x=infinity); value(%);
      
$$\lim_{x \rightarrow \infty} \frac{e^x}{x^3 + x - 1 + 0.2 e^x}$$

      5.
```

So the lines  $y = 0$  and  $y = 5$  are horizontal asymptotes as  $x \rightarrow -\infty$  and  $x \rightarrow \infty$ , respectively.


From the plot, there appears to be a local minimum between  $x = 1$  and  $x = 5$ . Its location and value can be pinpointed by setting the derivative to zero.

```
> Df:=diff(f,x);
> xmin:=fsolve(Df=0,x=1..5);
      xmin := 2.893289196
> ymin:=evalf(subs(x=xmin,f));
      ymin := 0.6073428968
```

Try the last command without the `evalf`. Notice that Maple does not automatically simplify after a `subs` command.

It is instructive to replot both  $f$  and its derivative on the same coordinate axes. We focus on the part of the graph of  $f$  over the interval  $0 \leq x \leq 15$ , since the local minimum of  $f$  is located there.

```
> plot({f,Df}, x=0..15, y=-2..6, discontin=true);
```



The scale for the  $y$ -coordinate is changed to more clearly display the graphs. Note that  $f$  decreases on the interval  $0.53 < x < 2.89$  and the derivative of  $f$

is negative on this same interval. Likewise,  $f$  increases on  $x > 2.89$  and the derivative of  $f$  is positive on this same interval. The graph of  $f$  flattens as  $x$  gets large. This corresponds to the observation that the graph of the derivative approaches  $y = 0$  when  $x$  gets large.


There also appears to be an inflection point (a point where the concavity of the graph switches) between  $x = 6$  and  $x = 10$ . Its location can be pinpointed by computing the second derivative (the derivative of the first derivative,  $Df$ ) and finding where it is zero.

```
> DDf:=diff(Df,x):
> xinfl:=fsolve(DDf=0,x=6..10);
      xinfl := 8.131398912
> yinfl:=evalf(subs(x=xinfl,f));
      yinfl := 2.775853638
```

So the point of inflection is approximately  $(8.131, 2.776)$ .

We now plot the expression  $f$  and its second derivative on the same coordinate axes.

```
> plot({f,DDf}, x=0..15, y=-1..5, discontin=true);
```



Notice that  $f$  is concave up and the second derivative is positive on the interval  $0.53 < x < 8.13$ . The graph of  $f$  is concave down and the second derivative is negative on the interval  $x > 8.13$ .

Also note that the graph of  $f$  is steepest where the first derivative is largest (see previous graph). This point,  $x = 8.13$ , is where the second derivative is zero.

## 5.4 Designer Polynomials

Here is an example that combines many of the Maple commands introduced up to this point. It is an extension of the ideas introduced in Example 1 of Section 3.1.

EXAMPLE: Find the coefficients of the cubic polynomial

$$f(x) = ax^3 + bx^2 + cx + d$$

that has a relative minimum at  $(-2, 0)$  and a relative maximum at  $(3, 4)$ . Plot the points and the polynomial.

SOLUTION: Four equations are needed for the four unknowns  $a$ ,  $b$ ,  $c$  and  $d$ . These equations will be labeled **eq1** through **eq4** for later reference. After defining  $f$  as a Maple function, the first two equations can be obtained from the information  $f(-2) = 0$  and  $f(3) = 4$ .

```
> f:=x->a*x^3+b*x^2+c*x+d;
      f := x → ax3 + bx2 + cx + d
> eq1:=f(-2)=0; eq2:=f(3)=4;
      eq1 := -8a + 4b - 2c + d = 0
      eq2 := 27a + 9b + 3c + d = 4
```

The remaining two equations can be obtained from the fact that the derivative of  $f$  must vanish at  $x = -2$  and  $x = 3$  (because  $f$  has a local minimum and maximum at these points).

```
> eq3:=D(f)(-2)=0; eq4:=D(f)(3)=0;
      eq3 := 12a - 4b + c = 0
      eq4 := 27a + 6b + c = 0
```

Now these four equations are solved for  $a$ ,  $b$ ,  $c$ , and  $d$ .

```
> sol:=solve({eq1,eq2,eq3,eq4},{a,b,c,d});
      sol := {a = -8/125, b = 12/125, c = 144/125, d = 176/125}
```


NOTE: In the `solve` command, Maple requires curly braces `{ }` or square brackets `[ ]` enclosing the equations `{eq1,eq2,eq3,eq4}` and the variables `{a,b,c,d}`. If the variables are enclosed in braces, then the output `sol` is also in the form of a set. Within the set, your solutions may appear in a different order. If the variables are enclosed in brackets, then the output `sol` is a list of lists and must be handled differently (which will not be discussed here).

To substitute the values of  $a$ ,  $b$ ,  $c$ ,  $d$  back into the original function, type

```
> g:=subs(sol,f(x));
      g := -8/125 x3 + 12/125 x2 + 144/125 x + 176/125
```

Finally, we plot the points and the graph of  $g$  to see it has a local minimum at  $(-2, 0)$  and a local maximum at  $(3, 4)$ .

```
> with(plots):
> p1:=plot([[-2,0],[3,4]], x=-4..5, style=point, symbol=box,
> symbolsize=24):
> p2:=plot(g, x=-4..5):
> display({p1,p2});
```




## 5.5 Absolute Extrema on an Interval

In this section, Maple is used to help find the absolute maximum and/or minimum (also called absolute extrema) of a function (or expression) on a (closed and bounded) interval. The method compares the values of the function at its critical points on the given interval with those at the endpoints of the interval. This is illustrated by the following example.

**EXAMPLE 1:** Find the absolute maximum and minimum of the expression  $f = \ln(x) - 4x^2 + x^3$  on the interval  $1 \leq x \leq 3$ .

**SOLUTION:** First, enter  $f$  as a function in Maple and plot it over the interval  $1 \leq x \leq 3$ .

```
> f:=x->ln(x)-4*x^2+x^3;
      f := x → ln(x) - 4x2 + x3
> plot(f, 1..3);
```



From the plot, it appears that the minimum of  $f$  is located at a critical point near  $x = 2.6$  and that the maximum of  $f$  is located at the left hand endpoint  $x = 1$ . To find the critical point, set the derivative of  $f$  equal to zero and solve.

```
> xmin:=fsolve(D(f)(x)=0,x=2..3);
      xmin := 2.618033989
```



The values of  $f$  at the critical point  $x = x_{min}$  and the endpoints  $x = 1$  and  $x = 3$  are

```
> f(xmin), f(1), f(3.);
      -8.50971230, -3, -7.90138771
```

So, the maximum of  $f$  is  $-3$  at  $x = 1$  and the minimum of  $f$  is  $-8.51$  at  $x = 2.62$ .


If the interval is open (does not contain its endpoints) or if the interval is infinite, then there may not be an absolute maximum or minimum.

EXAMPLE 2: Find any absolute maxima or minima of the expression

$f = \frac{x^4 + 2x + 3}{x^2}$  on the interval  $x > 0$ .

SOLUTION: Enter the expression and plot it.

```
> f:=(x^4+2*x+3)/x^2;
      f :=  $\frac{x^4 + 2x + 3}{x^2}$ 
> plot(f, x=0..10, y=-50..100);
```



The plot reveals an absolute minimum at a critical point near  $x = 1.5$ . However, this expression has no absolute maximum, since  $f$  approaches infinity as  $x \rightarrow 0^+$  or  $x \rightarrow \infty$ . The minimum can be found by setting the derivative,  $Df := \text{diff}(f, x)$ , equal to zero and solving:  $\text{fsolve}(Df=0, x=1..2)$ . The minimum of  $f$  is approximately 4.9087, occurring at  $x = 1.4526$ . Try it.


## 5.6 The Most Economical Tin Can

This section presents an example of an applied max/min problem that cannot be easily solved by hand. The discussion will focus on the problem solving strategy and the role of Maple in that strategy rather than on the Maple syntax itself.

For an applied max/min problem, you must

- i. translate the problem into mathematics, (Maple will not help you here.)
- ii. solve the resulting calculus problem, (This is where Maple can help.)  
and finally,
- iii. interpret the results. (Again Maple cannot help.)

EXAMPLE: Consider a cylindrical tin can (shown below) which is to be constructed by joining the ends of a rectangular piece of material to form the curved side and then attaching circular pieces to form the top and bottom. There are seams around the perimeter of the top and bottom, and there is one seam down the side surface (where the ends of the rectangle join together).



Suppose the volume of the can is 750 cubic centimeters. Also suppose that the cost of the material is \$1.10 per square meter and the cost of the seam is \$0.25 per meter. Find the dimensions of the can that will minimize its cost. Find the minimum cost.

SOLUTION: Let  $r$  be the radius of the base of the can and  $h$  be the height of the can. The surface area of the can is

$$> \text{area} := 2 * \text{Pi} * r^2 + 2 * \text{Pi} * r * h;$$

$$\text{area} := 2 \pi r^2 + 2 \pi r h$$

The first term on the right represents the area of the top and bottom. The second term represents the area of the side (which has the same area as a rectangle of length  $2\pi r$  and height  $h$ ).

The total length of the seams is

$$> \text{len} := 4 * \text{Pi} * r + h;$$

$$\text{len} := 4 \pi r + h$$

We will use centimeters and cents as our units. The cost of the material is  $a = 0.011$  cents per square centimeter (110 cents divided by the 10000 square centimeters in a square meter). The seaming cost is  $b = 0.25$  cents per centimeter. So the total cost of the can is

$$> \text{a} := 0.011; \text{b} := 0.25; \text{cost} := \text{a} * \text{area} + \text{b} * \text{len};$$

$$a := 0.011$$

$$b := 0.25$$

$$\text{cost} := 0.022 \pi r^2 + 0.022 \pi r h + 1.00 \pi r + 0.25 h$$

This cost expression has both  $r$  and  $h$  as variables. However, there is a constraint between the variables which we can use to eliminate  $h$ , namely that the volume is 750 cubic centimeters. Since the volume is

$$\text{volume} = \text{base} \times \text{height} = \pi r^2 h = 750$$

we have

$$> \text{h} := 750 / (\text{Pi} * r^2);$$

$$h := \frac{750}{\pi r^2}$$


The cost is now

```
> cost;
```

$$0.022 \pi r^2 + \frac{16.500}{r} + 1.00 \pi r + \frac{187.50}{\pi r^2}$$

which must be minimized over the interval  $r > 0$ . Its plot is

```
> plot(cost,r=1..10);
```



Observe that a minimum occurs somewhere between  $r = 2$  and  $r = 6$ . To find it, set its derivative equal to zero and solve for  $r$ :

```
> Dc:=diff(cost,r); rmin:=fsolve(Dc=0,r=2..6);
```

$$Dc := 0.044 \pi r - \frac{16.500}{r^2} + 1.00 \pi - \frac{375.00}{\pi r^3}$$

$$rmin := 3.666752590$$

The answer is the best value for the radius (in centimeters). Substituting this value into the expression for  $h$  yields

```
> hmin:=evalf(subs(r=rmin,h));
```

$$hmin := 17.75612426$$

So the minimum cost (in cents) is:

```
> cmin:=evalf(subs(r=rmin,cost));
```

$$cmin := 21.38762546$$

## 5.7 Summary

**Related Rates:** The procedure for solving related rates problems can be outlined as follows:

1. Determine the quantities (say  $u$  and  $v$ ) that are changing with time (or another parameter).
2. Find an equation that relates the quantities. (This may involve known formulas from geometry or physics, etc.)

3. Enter the equation (say `eq`) into Maple, writing the quantities as functions of time. (Here  $u(t)$  and  $v(t)$ .)
4. Take the derivative of the equation with respect to time.  
`deq:=diff(eq,t);`
5. Solve the resulting equation for the desired implicit derivative.  
`vrate:=solve(deq,diff(v(t),t));`
6. Find numeric values for each of the quantities in the resulting solution, possibly by using the equation from step 3, and substitute them into the result.  
`vrate0:=subs(u(t)=u0, diff(u(t),t)=urate0, v(t)=v0, vrate);`

**Graphical Analysis of Expressions:** Use the following commands to analyze an expression labeled `expr`:

- Plot:
  - > `f:=expr;`
  - > `plot(f,x);`
- $x$ -Intercepts: From the plot, determine an interval  $[a, b]$  containing one  $x$ -intercept.
  - > `fsolve(f=0,x=a..b);`
 Repeat for each  $x$ -intercept.
- $y$ -Intercept:
  - > `subs(x=0,f);`
- Local maxima and minima: From the plot determine an interval  $[c, d]$  containing a maximum or minimum.
  - > `Df:=diff(f,x);`
  - > `x0:=fsolve(Df=0,x=c..d);`
  - > `y0:=evalf(subs(x=x0,f));`
 Repeat for each maximum or minimum.
- Inflection Points: From the plot determine an interval  $[r, s]$  containing an inflection point.
  - > `DDf:=diff(Df,x);`
  - > `xinfl:=fsolve(DDf=0,x=r..s);`
  - > `yinfl:=evalf(subs(x=xinfl,f));`
 Repeat for each inflection point.

- Vertical Asymptotes: From the plot determine an interval  $[p, q]$  containing a vertical asymptote. If necessary, use `f:=simplify(f)`; to write `f` as a single fraction, rather than a sum of fractions.

```
> vasymp:=fsolve(denom(f)=0,x=p..q);
```

Repeat for each vertical asymptote.

- Horizontal asymptote:
  - > `Limit(f,x=-infinity); value(%)`;
  - > `Limit(f,x=infinity); value(%)`;

**Graphical Analysis of Arrow-Defined Functions:** Use the following commands to analyze a function `f` defined from an expression `expr` by:

```
> f:=x->expr;
```

- Plot:
  - > `plot(f(x),x)`;
- $x$ -Intercepts: From the plot, determine an interval  $[a, b]$  containing one  $x$ -intercept.

```
> fsolve(f(x)=0,x=a..b);
```

Repeat for each  $x$ -intercept.

- $y$ -Intercept:
  - > `f(0)`;

- Local maxima and minima: From the plot determine an interval  $[c, d]$  containing a maximum or minimum.

```
> Df:=D(f);
> x0:=fsolve(Df(x)=0,x=c..d);
> y0:=evalf(f(x0));
```

Repeat for each maximum or minimum.

- Inflection Points: From the plot determine an interval  $[r, s]$  containing an inflection point.

```
> DDf:=D(Df);
> xinfl:=fsolve(DDf(x)=0,x=r..s);
> yinfl:=evalf(f(xinfl));
```

Repeat for each inflection point.

- Vertical Asymptotes: From the plot determine an interval  $[p, q]$  containing a vertical asymptote. If necessary, use `simplify(f(x))`; to write `f(x)` as a single fraction, rather than a sum of fractions.

```
> vasymp:=fsolve(denom(f(x))=0,x=p..q);
```

Repeat for each vertical asymptote.

- Horizontal asymptote:
  - > `Limit(f(x),x=-infinity); value(%);`
  - > `Limit(f(x),x=infinity); value(%);`

#### Maximizing or Minimizing on an Interval:

- Define and plot the Maple function on the interval  $a \leq x \leq b$ .
  - > `f:=x->expr;`
  - > `plot(f(x), x=a..b);`
- Use the plot and the `fsolve` command to find the  $x$ -coordinates of the extrema in the interior of this interval.
  - > `Df:=D(f);`
  - > `xcrit:=fsolve(Df(x)=0,x=c..d);`

If there is more than one critical point, repeat this and give them different labels.


- Evaluate the function at each of these critical points and the two endpoints:
  - > `f(xcrit);`
  - > `f(a);`
  - > `f(b);`
- Compare these. The largest is the absolute maximum; the smallest is the absolute minimum.

## 5.8 Exercises

1. When a ray of light hits the surface of a lake, the beam is bent. The equation that governs this effect is due to Willebrod Snell (1591–1626). He noted that

$$\frac{\sin(\alpha)}{\sin(\beta)} = 1.33$$

where 1.33 is the index of refraction of water (relative to air) and  $\alpha$  and  $\beta$  are the angle of incidence and angle of refraction, respectively, measured from a line perpendicular to the surface of the lake. Since the earth rotates once in 24 hours, as the sun rises, the angle of incidence of the sunlight decreases at  $\pi/12$  radians per hour. When  $\alpha = \pi/3$ , how fast does a fish see the sun rise?



2. Plot the expression  $f = .25x^4 - .31x^3 - .12x^2 + .25x - 0.4$ . You may need to adjust the plot ranges to include all relevant aspects of the graph. Find the local maxima and minima and the intervals where  $f$  is increasing and decreasing.
3. Find the cubic  $ax^3 + bx^2 + cx + d$  that has a local minimum at  $(-1, -2)$  and a local maximum at  $(4, 4)$ . Plot it.
4. Find the cubic polynomial that passes through the points  $(1, -1)$ ,  $(5, 4)$ , and  $(6, 4)$  and whose derivative at  $x = 3$  is  $3.2$ . Plot it.
5. Find the cubic polynomial with a local maximum at  $(2, 7)$  and a local minimum at  $(5, 2)$ . Plot it.
6. Plot one or more informative graphs of  $y = \frac{|x^3 - 5.8x^2 + 11.5x - 5.4|}{x^3 - 3.1x^2 - 3.2x + 4.21}$ . Determine all intercepts, asymptotes and local extrema.
7. Graph the function  $y = \frac{x \ln(x)}{x^2 + x + 4}$  for  $x > 0$ . Find all critical points and inflection points. Find intervals where the function is increasing, decreasing, concave up, and concave down.
8. For each of the following expressions, locate the horizontal and vertical asymptotes, the local extrema and the inflection points. Then graph the given expression, its derivative and second derivative in one plot using different colors. On each of the intervals where the expression is increasing (decreasing), is the derivative positive or negative? Do a similar comparison for concavity of the expression and the sign of the second derivative.

(a)  $\frac{10x^2 + 4x + 1}{5x - 1}$

(b)  $\frac{3x^4 + 2x + 1}{x^4 - 3x^2 + 1}$

(c)  $\frac{3e^x}{e^x - x^6}$  (Type  $e^x$  as **exp(x)**.)

- (d)  $x^2 + 4 \sin(x) \cos(x)$   
 (e)  $\ln(x) + 7x^2 - x^3$  for  $x > 0$

9. (This is a warm-up for the next exercise.) A plot of land has straight southern, eastern and western boundaries and an irregular northern boundary. The goal of this problem is to draw a map of the boundary of this plot of land using cubic polynomials (called a cubic spline). Since the southern and western boundaries are straight, they will be represented by the  $x$ -axis and the  $y$ -axis respectively. North-south measurements (represented by the variable  $y$ ) are taken and placed in the following table (each unit represents 100 feet).

$x$ -value	0.0	1.0	2.0	3.0	4.0
$y$ -value	3.2	1.5	2.6	3.2	2.7

The first task is to find the graph of the quadratic polynomial  $p(x) = ax^2 + bx + c$  (a parabola) that contains the first three data points. This is easily accomplished by defining  $p$  as a function and solving the equations  $p(0) = 3.2$ ,  $p(1) = 1.5$ , and  $p(2) = 2.6$  for the unknowns  $a$ ,  $b$ , and  $c$ .

The next task is to find the graph of the cubic polynomial  $q(x) = dx^3 + ex^2 + fx + g$  that contains the last three data points and satisfies  $p'(2) = q'(2)$ . This last equation ensures that the graphs of  $p$  and  $q$  have the same slope at  $x = 2$ . To do this, enter  $q$  as a function (if necessary, unassign any previously used labels) and solve the equations  $q(2) = 2.6$ ,  $q(3) = 3.2$ ,  $q(4) = 2.7$ , and  $D(p)(2) = D(q)(2)$  for  $d$ ,  $e$ ,  $f$  and  $g$ .

Since  $p$  and  $q$  are plotted over different intervals, graph them using the `display` command, as explained in Section 2.2

10. *Smooth off the southern boundary of Texas.* In Exercise 2 in Chapter 3, the boundary of the state of Texas was drawn using line segments. The point of this problem is to use parabolas and cubics (called a cubic spline) to smooth out the part of the southern boundary of the state formed by the Rio Grande River. Enter the relevant Rio Grande data:  
`> rio:=[0,0], [1,-1.1], [2,-2.5], [3,-2.9], [4,-2.3],`  
`> [5,-2.8], [6,-4.4], [7,-5.8], [8,-6.1]]:`


Here the origin is the westernmost corner of Texas (near El Paso) and the  $x$ -axis is the extension of the east-west border between New Mexico and Texas. Each unit represents approximately 69 miles.

To find functions that smooth out the Rio Grande, proceed as in Exercise 9. First, find the parabola  $p(x)$  that passes through the three data points  $[0, 0]$ ,  $[1, -1.1]$ , and  $[2, -2.5]$ . Then find a cubic polynomial  $q$  that passes through the next triplet of data points  $[2, -2.5]$ ,  $[3, -2.9]$ ,  $[4, -2.3]$  and further satisfies the equation  $p'(2) = q'(2)$  (so that the slopes of the graphs of  $p$  and  $q$  at  $x = 2$  are the same). In the same manner, find cubics  $r$  and  $s$  for the triplets  $[4, -2.3]$ ,  $[5, -2.8]$ ,  $[6, -4.4]$  and  $[6, -4.4]$ ,  $[7, -5.8]$ ,  $[8, -6.1]$




so that  $q'(4) = r'(4)$  and  $r'(6) = s'(6)$ . Then plot  $p(x)$  for  $0 \leq x \leq 2$ ;  $q(x)$  for  $2 \leq x \leq 4$ ;  $r(x)$  for  $4 \leq x \leq 6$  and  $s(x)$  for  $6 \leq x \leq 8$  and display them on the same coordinate axes.


11. In Section 5.6 we found the dimensions of a tin can which minimized its cost. Repeat that problem, but assume you need to pay for each full square of material used to cut out the circular ends of the cylinder.
12. A metal box with a square base and no top holds 900 cubic centimeters. It is formed by folding up the sides of the flattened pattern pictured here and seaming up the four sides. The material for the box costs \$1.00 per square meter and the cost to seam the sides is 6 cents per meter. Find the dimensions of the box that costs the least to produce assuming you do not need to pay for the metal squares which have been cut out of the corners.



13. Repeat Exercise 12 but assume you do need to pay for the metal squares which have been cut out of the corners.
14. A pipeline is to be constructed to connect a station on the shore of a straight section of coast line to a drilling rig that lies 7 kilometers down the coast and 4 kilometers out at sea. Find the minimum cost to construct the pipeline, given that the pipeline costs 4.6 million dollars per kilometer to lay under water and 2.4 million dollars per kilometer to lay along shore.
15. A rectangular movie theater is 100 feet long (from the front screen to the back). The top and bottom of its screen are 40 feet and 15 feet from the floor, respectively. Find the position in the theater with the largest viewing angle.  
 HINT: Arrange a coordinate system with the origin at the floor directly under the screen. You are asked to find the position  $x$  where the angle  $\alpha - \beta$  is the largest (see figure). Instead of maximizing  $\alpha - \beta$ , it is easier to maximize  $\tan(\alpha - \beta)$ . This will lead to the same optimum value of  $x$ , since  $\tan(\theta)$  is an increasing function on  $-\pi/2 < \theta < \pi/2$ . The subtraction formula for tangent will be needed. This may be found by executing `expand(tan(a-b))`;



16. Repeat Exercise 15, but this time assume that the floor of the theater has a gentle parabolic slope given by the parabola  $y = 0.001x^2$  for  $0 \leq x \leq 100$ . (The origin is located on the floor directly under the screen.)
17. Repeat Exercise 16, but assume that the floor of the theater has a steeper parabolic slope given by the formula  $y = 0.004x^2$ .
18. Two hallways of width  $a$  and  $b$  intersect at right angles. What is the length of the longest rigid rod that can be pushed on the floor around the corner of the intersection of these two halls? Carefully explain your steps. HINT: First try it for  $a = 3$  and  $b = 4$ .




19. Find the point on the graph of  $y = x^2 \ln(x)$  closest to the point  $(2, 7)$ .  
HINT: Minimize the function that describes the square of the distance between the point  $(2, 7)$  and a typical point  $(x, x^2 \ln(x))$  on the graph.
20. Find the point on the parabola  $y = \frac{x^2}{3} - 1$  closest to the point  $(-1, 5)$ .
21. The point of this problem is to derive Snell's Law:

$$\frac{\sin(\theta_1)}{\sin(\theta_2)} = \frac{v_1}{v_2}$$


where  $\theta_1$  and  $v_1$  are the angle of incidence and the velocity of light in the first medium (e.g. air),  $\theta_2$  and  $v_2$  are the angle of refraction and the velocity of light in the second medium (e.g. water) as shown in the figure.

(This equation can be used to calculate  $\theta_2$  provided  $\theta_1$ ,  $v_1$ , and  $v_2$  are known.) We know from physics that a ray of light travels from a point  $A$  in the air to a point  $B$  in the water via a path  $ACB$  that minimizes the time taken where  $C$  is a point on the surface of the water.



HINT: Write a formula for the time it takes to travel from  $A$  to  $C$  to  $B$ . Then vary  $C$  to minimize this formula. In the figure,  $a$ ,  $b$ ,  $c$  and  $d$  are fixed,  $x$  varies. Use Maple to compute a derivative, if you wish. Also express  $\sin \theta_1$  and  $\sin \theta_2$  in terms of  $a$ ,  $b$ ,  $c$ ,  $d$  and  $x$ . The rest of the solution is more easily done by hand.

22. Now suppose the medium in Exercise 21 has a curved surface (such as the surface of a glass lens). To be more specific, suppose a beam of light follows the line with the equation  $y = 3x + 2$  and strikes a piece of glass whose outside boundary is given by the equation  $y = x^2$ . (See the figure.) Find the equation of the line that represents the refracted light beam as it travels through the glass. Assume that  $v_1/v_2$  is 1.5.



HINT: First, use Maple to find the point of intersection between the light beam and the parabolic surface of the glass. At this point compute the angle between the normal to the parabolic surface and the incoming light beam. This angle is the angle  $\theta_1$  of Exercise 21. Snell's law can now be used to calculate  $\theta_2$  and the slope of the refracted light can then be determined.

23. Suppose a glass lens is formed by the intersection of the interiors of the following two ellipses:

$$(x - 4)^2 + \frac{y^2}{2} = 26 \quad \text{and} \quad (x + 4)^2 + \frac{y^2}{2} = 26.$$

A light beam strikes this lens from the left along the horizontal line  $y = 3$ . Find the equations of the line segments that describe the trajectories of the refracted light beam as it passes through the lens and as it passes through the air on the right side of the lens. Find the location where the refracted light beam hits the positive  $x$ -axis.

